

# Pascal

## 04 - Manipulação de Strings

---

Marcos Roberto Ribeiro



Instituto Federal Minas Gerais - Campus Bambuí

2018

- O **String** é um tipo especial de vetor adequado para trabalhar com caracteres. As principais vantagens dos **Strings** são o tamanho dinâmico e a existência de diversas rotinas prontas para manipulação de strings;
- O tamanho dinâmico significa que um **String** não tem um tamanho fixo podendo aumentar ou diminuir conforme a necessidade;
- Outra propriedade interessante do tipo **String** é a possibilidade de concatená-lo através do operador  $+$ .

- Dentre as rotinas de manipulação de *Strings* podemos destacar:
  - Length()** Obtém o número de caracteres presentes na **String**;
  - Pos()** Localiza a posição de uma **String** dentro de outra **String**;
  - Copy()** Copia um trecho de caracteres de uma **String**;
  - Delete()** Apaga um trecho de caracteres de uma **String**;
  - Insert()** Insere um trecho de caracteres dentro de uma string;
  - Val()** Conversão de *String* para *Real* ou **Integer** com verificação;
  - Str()** Conversão de *Real* ou **Integer** para *String*;

## A Rotina Length()

- A função **Length(Texto: String): Integer** recebe um parâmetro **Texto** do tipo **String** e retorna o número de caracteres presentes no **String**;
- A função **Pos(SubTexto, Texto: String): Integer** localiza a posição de uma **String** dentro de outra **String**. O parâmetro **SubTexto** é o **String** a ser procurado e o parâmetro **Texto** é onde será realizada a busca. A função **Pos()** retorna a posição do primeiro caractere de **SubTexto** ou 0 (zero) se não existir nenhuma ocorrência da **String** procurada;

## Exemplo com Length() e Pos() I

```
var
  NomeCompleto, PrimeiroNome, NomeMeio: String;
  Palavra, Sobrenome, Iniciais: String;
  Cont, Posicao: Integer;
BEGIN
  {Obtém os nomes}
  Write('Entre com o seu nome: ');
  ReadLn(PrimeiroNome);
  Write('Entre com o seu nome do meio: ');
  ReadLn(NomeMeio);
  Write('Entre com o seu sobrenome: ');
  ReadLn(Sobrenome);

  {Concatena os nomes para formar o nome completo}
  NomeCompleto:= PrimeiroNome + ' ' + NomeMeio + ' ' + Sobrenome;
  WriteLn('Seu nome completo é ', NomeCompleto);
  {Obtém as iniciais a partir do nome completo}
  Iniciais:= '';
```

## Exemplo com Length() e Pos() II

```
for Cont:= 1 to Length(NomeCompleto) do begin
    {Testa se o caractere é uma inicial}
    if (NomeCompleto[Cont] <> ' ') and ((Cont = 1) or (NomeCompleto[Cont - 1] = '
↪ ')) then begin
        Iniciais:= Iniciais + NomeCompleto[Cont];
    end;
end;
WriteLn('Suas iniciais são ', iniciais);

Write('Informe uma palavra a ser buscada no seu nome: ');
ReadLn(Palavra);
Posicao:= Pos(Palavra, NomeCompleto);
if (Posicao > 0) then begin
    WriteLn('A palavra foi encontrada na posição ', Posicao);
end else begin
    WriteLn('A palavra não foi encontrada');
end;
END.
```

## As Rotinas `Copy()`, `Insert()` e `Delete()`

- A função `Copy(Texto: String; Ini, Quant: Integer): String` retorna uma parte da `String Texto` contendo `Quant` caracteres, começando pela posição `Ini`;
- O procedimento `Insert(TextoIns: String; var Texto: String; Pos: Integer)` insere `TextoIns` na posição `Pos` de `Texto`;
- O procedimento `Delete(var Texto: String; Ini, Quant: Integer)` apaga `Quant` caracteres de `Texto` começando pela posição `Ini`.

## Exemplo utilizando Copy(), Insert() e Delete()

```
var
    Telefone, DDD, Digitos: String;
BEGIN
    {Obtém o DDD e telefone}
    WriteLn('Informe seu DDD e telefone (apenas os números)');
    Write('DDD: ');
    ReadLn(DDD);
    Write('Telefone: ');
    ReadLn(Telefone);
    {Copia os quatro últimos dígitos do telefone}
    Digitos:= Copy(Telefone, 5, 4);
    WriteLn('Os quatro digitos finais são: ', Digitos);
    {Insere - na quinta posição do telefone}
    Insert('-', Telefone, 5);
    {Concatena o telefone com o DDD}
    Telefone:= '(' + DDD + ')' + Telefone;
    WriteLn('Seu telefone com DDD é: ', Telefone);
    Delete(Telefone, 1, 4); {Remove o DDD}
    WriteLn('Seu telefone sem DDD é: ', Telefone);
END.
```



## As Rotinas Val() e Str()

- O procedimento **Val(Texto: String; var Num: Real (ou Integer); var Erro: Integer)** tenta converter a **String Texto** para o número **Num**. Se a conversão ocorrer normalmente **Erro** recebe 0 (zero);
- O procedimento **Str(Num: Real (ou Integer); var Texto: String)** converte o valor numérico de **Num** para a **String Texto**.

## Exemplo utilizando Val() e Str()

```
var
    Num_String, Mensagem: String;
    Num: Real;
    Erro: Integer;
BEGIN
    Write('Informe um número: ');
    ReadLn(Num_String);
    Val(Num_String, Num, Erro); {Tenta converter "Num"}
    {Verifica se ocorreu erro na conversão}
    if (Erro <> 0) then begin
        WriteLn('Você informou um número inválido!');
    end else begin
        Str(Num:0:2, Num_String); {Converte número para String}
        Mensagem := 'O quadrado de ' + Num_String + ' é ';
        Num := Num * Num;
        Str(Num:0:2, Num_String);
        Mensagem := Mensagem + Num_String;
        WriteLn(Mensagem);
    end;
END.
```

# Referências I



Canneyt, M. V.

**Free pascal reference guide.**

<http://www.freepascal.org/docs-html/ref/ref.html>.



Evaristo, J. (1999).

***Programando com Pascal.***

Book Express, Rio de Janeiro, 2 edition.



Manzano, J. A. N. G. and Yamatumi, W. Y. (2001).

***Programando em turbo pascal 7.0 e free pascal compiler.***

Érica, São Paulo, 9 edition.